

/* MISS-MANQUER via une interlingua */

/* Version zéro */

/*

pierre lui manque

she misses pierre

he misses pierre

mary misses pierre

pierre manque à marie

they miss mary

mary leur manque

*/

:- style_check(-singleton).

```
translate(fr,Phrase,Traduction,Interlingua) :-  
    analyse(Phrase,[],Interlingua),  
    parse(Traduction,[],Interlingua),  
    writelist(Traduction).
```

```
translate(gb,Sentence,Translation,Interlingua) :-  
    parse(Sentence,[],Interlingua),  
    analyse(Translation,[],Interlingua),  
    writelist(Translation).
```

/* la grammaire française */

```
analyse(P0,[],m(misser(M_exp),
                missee(M_pat)
                )
        ) :-
    phr(P0,[],[M_exp,M_pat]) .
```

```
phr(P0,P2,[M_exp,M_pat]) :-
    gn(P0,P1,M_pat,P,N,Type,sujet) ,
    gv(P1,P2,M_exp,P,N).
```

```
gn(P0,P1,N,Personne,Nombre,plein,Fonction) :-
    propre(P0,P1,N,Personne,Genre,Nombre,Fonction).
```

```
gn(P0,P1,Pr,Personne,Nombre,pronominal,Fonction) :-
    pronom(P0,P1,Pr,Personne,Nombre,Genre,Fonction).
```

```
gv(P0,P2,M_exp,Personne,Nombre) :-
    v_manquer(P0,[à | P1],Personne,Nombre),
    gn(P1,P2,M_exp,P,N,plein,_).
```

```
gv(P0,P2,M_exp,Pe,No) :-
    gn(P0,P1,M_exp,P,N,pronominal,preverbe),
    v_manquer(P1,P2,Pe,No).
```

```
/* English grammar */
```

```
parse(P0,[],m(misser(M_exp),  
             missee(M_pat)  
             )  
      ) :-  
s(P0,[],[M_exp,M_pat]) .
```

```
s(P0,P2,[M_exp,M_pat]) :-  
  np(P0,P1,M_exp,P,N,subject) ,  
  vp(P1,P2,M_pat,P,N).
```

```
np(P0,P1,N,Person,Number,Function) :-  
  proper(P0,P1,N,Person,Gender,Number,Function).
```

```
np(P0,P1,Pr,Person,Number,Function) :-  
  pronoun(P0,P1,Pr,Person,Number,Gender,Function).
```

```
vp(P0,P2,M_pat,Person,Number) :-  
  v_miss(P0,P1,Person,Number),  
  np(P1,P2,M_pat,P,N,non_subject).
```

/* le lexique français */

v_manquer([manque | X],X,1,sg).
v_manquer([manques | X],X,2,sg).
v_manquer([manque | X],X,3,sg).
v_manquer([manquons | X],X,1,pl).
v_manquer([manquez | X],X,2,pl).
v_manquer([manquent | X],X,3,pl).

propre([pierre | X],X,n(pierre),3,masc,sg,_).
propre([paul | X],X,n(paul),3,masc,sg,_).
propre([jean | X],X,n(jean),3,masc,sg,_).
propre([marie | X],X,n(marie),3,fem,sg,_).
propre([peter | X],X,n(peter),3,masc,sg,_).
propre([john | X],X,n(john),3,masc,sg,_).
propre([mary | X],X,n(mary),3,fem,sg,_).

pronom([je | X],X,pro(p1,sg,_),1,sg,_,sujet).
pronom([tu | X],X,pro(p2,sg,_),2,sg,_,sujet).
pronom([il | X],X,pro(p3,sg,masc),3,sg,masc,sujet).
pronom([elle | X],X,pro(p3,sg,fem),3,sg,fem,sujet).
pronom([nous | X],X,pro(p1,pl,_),1,pl,_,_).
pronom([vous | X],X,pro(p2,pl,_),2,pl,_,_).
pronom([ils | X],X,pro(p3,pl,masc),3,pl,masc,sujet).
pronom([elles | X],X,pro(p3,pl,fem),3,pl,fem,sujet).

pronom([me | X],X,pro(p1,sg,_),1,sg,_,preverbe).
pronom([te | X],X,pro(p2,sg,_),2,sg,_,preverbe).
pronom([lui | X],X,pro(p3,sg,_),3,sg,_,preverbe).
pronom([leur | X],X,pro(p3,pl,_),3,pl,_,preverbe).

/* English lexicon */

```
v_miss([miss | X],X,1,sg).
v_miss([miss | X],X,2,sg).
v_miss([misses | X],X,3,sg).
v_miss([miss | X],X,1,pl).
v_miss([miss | X],X,2,pl).
v_miss([miss | X],X,3,pl).
```

```
proper([pierre | X],X,n(pierre),3,masc,sg,_).
proper([paul | X],X,n(paul),3,masc,sg,_).
proper([jean | X],X,n(jean),3,masc,sg,_).
proper([marie | X],X,n(marie),3,fem,sg,_).
proper([peter | X],X,n(peter),3,masc,sg,_).
proper([john | X],X,n(john),3,masc,sg,_).
proper([mary | X],X,n(mary),3,fem,sg,_).
```

```
pronoun([i | X],X,pro(p1,sg,_),1,sg,_,subject).
pronoun([you | X],X,pro(p2,_,_),2,_,_,_).
pronoun([he | X],X,pro(p3,sg,masc),3,sg,masc,subject).
pronoun([she | X],X,pro(p3,sg,fem),3,sg,fem,subject).
pronoun([we | X],X,pro(p1,pl,_),1,pl,_,subject).
pronoun([they | X],X,pro(p3,pl,_),3,pl,_,subject).
pronoun([me | X],X,pro(p1,sg,_),1,sg,_,non_subject).
pronoun([us | X],X,pro(p1,pl,_),1,pl,_,non_subject).
pronoun([him | X],X,pro(p3,sg,masc),3,sg,masc,non_subject).
pronoun([her | X],X,pro(p3,sg,fem),3,sg,fem,non_subject).
pronoun([them | X],X,pro(p3,pl,_),3,pl,_,non_subject).
```

```
/* OUTPUT */
```

```
writelist([H | T]) :- write(H),tab(1),writelist(T).
writelist([]):- nl
```